

Guía del desarrollador

ProRealTime.com
La referencia en flujo Tiempo Real

ProBacktest 1.0b
edición 2004

SUMARIO

Presentación de ProBacktest	2
Instrucciones específicas a ProBacktest.....	3
Simulación de la transmisión de órdenes	3
Simulación de stop-loss gestionados por programación	7
Estrategias de backtest	8
Variables de estado.....	10
Variables de seguimiento de posición.....	12
Acceso a las barras relativas a la última orden.....	14
Ejemplos.....	16
PennyStocks	16
Entrada optimizada.....	18
Tendencia detectada sobre las cotizaciones	20
Sell in may and go away !.....	21
Intradía 2 barras.....	22

Presentación de ProBacktest

ProBacktest es la herramienta de ProRealTime dedicada al « backtesting ». Su sintaxis es análoga a la de ProBuilder. Ello le confiere una compatibilidad que le permite, por ejemplo, poner condiciones sobre los indicadores creados con ProBuilder.

Se trata de un lenguaje de tipo BASIC con el cual usted podrá construir sus sistemas de « trading » automatizados en vista a aplicarlos a cualesquiera de los parámetros de ProRealTime. Para cada barra presente en la interfaz gráfica, estos parámetros son:

- la cotización de apertura
- la cotización de cierre
- la cotización más alta
- la cotización más baja
- el número de títulos cambiados.

Un aspecto esencial de ProBacktest es la representación, mediante flechas en la pantalla, de los niveles de compra/venta generados por su sistema. A partir de ahí, ProBacktest calcula la « equity curve », la cual le indica la performance de su sistema. La curva de equidad aparece entonces como un indicador por encima de las cotizaciones.

ProBacktest hace una valoración de los datos de cada una de las barras de cotizaciones, desde la más anciana hasta la más reciente, y ejecuta la fórmula desarrollada concerniente la determinación de las órdenes eventuales a simular sobre cada barra.

Instrucciones específicas a ProBacktest

Simulación de la transmisión de órdenes

Una orden es una instrucción que precisa: el sentido de la transacción a simular, el número de títulos en juego, las modalidades de ejecución, así como la fecha de ejecución. A veces, según el contexto, puede omitirse alguno de estos parámetros.

Sentido de la transacción

Buy, sell

BUY *count* **SHARES** (AT MARKET | AT price LIMIT | AT price STOP)

Esta instrucción entra en el mercado a la compra. Si la cartera de backtest es corta (o short) en el momento en el cual la orden debe ejecutarse, las posiciones cortas se ejecutan y la posición resultante corresponde a *count* títulos largos. Si la cartera es ya larga (posiciones de compra) en el momento en que la orden debe ser ejecutada, se aplican las reglas piramidales según los cánones definidos por usted en la rúbrica Money Management.

SELL [*count* **SHARES**] (AT MARKET | AT price LIMIT | AT price STOP)

Esta instrucción sale las posiciones largas del mercado. Si la cartera de backtest es líquido o corto en el momento de ejecución de la orden, ninguna acción tiene lugar. Por el contrario, si la cartera es largo, las posiciones *count* son ejecutadas. Lo mismo cuando *count* es superior al número de títulos que se posee: no se pasa a corto sobre el mercado, sólo se venden las posiciones largas. Si *count* se omite, todos los títulos que se poseen a la compra quedan vendidos.

Sellshort, exitshort

SELLSHORT *count* **SHARES** (AT MARKET | AT price LIMIT | AT price STOP)

Esta instrucción entra a la venta. Si la cartera de backtest es larga en el momento de ejecutarse la orden, las posiciones largas se efectúan y la posición resultante corresponde a *count* títulos cortos. Si la cartera es ya corto en el citado momento, se aplican las reglas piramidales según los criterios definidos por usted en la rúbrica Money Management.

EXITSHORT [*count* **SHARES**] (AT MARKET | AT price LIMIT | AT price STOP)

Esta instrucción hace salir las posiciones cortas del mercado. Si la cartera de backtest es líquida o larga en el momento en que la orden debe ejecutarse, no ocurre nada. Si es corto, las posiciones *count* se compran. Incluso cuando *count* es superior al número de títulos vendidos, no se pasa largo sobre el mercado, por el contrario las posiciones cortas se compran. Si se omite *count*, todos los títulos puestos a la venta al descubierto se compran.

Nombre de los títulos

El nombre de los títulos puede expresarse en unidades de cash, en fracciones de capital, o en fracciones de liquidez disponibles.

- **SHARES** nombre de los títulos
- **CASH** total de la transacción en unidades de cash (por ejemplo €ou \$)
- **%CAPITAL** total de la transacción en fracciones de capital
- **%LIQUIDITY** total de la transacción en fracciones de liquidez

Ejemplo Cash

REM Compra por 1000€ (por un título cotizado en euros)
BUY 1000 Cash AT MARKET

Ejemplo %Capital

REM Compra por 70% de la valorización de la cartera
BUY 70 %Capital AT MARKET

Ejemplo %Liquidity

REM Compra por 40% de la liquidez disponible
BUY 40 %Liquidity AT MARKET

Nota:

Cuando se trata de órdenes de salida del mercado (*SELL*, *EXITSHORT*), el nombre de los títulos es facultativo.

REM Venta de todas las posiciones largas
SELL AT MARKET

Modalidades de ejecución

Existen tres tipos de órdenes: al precio del mercado, al mejor precio límite y a nivel de arranque.

- **AT MARKET** al precio del mercado
- **AT price LIMIT** al mejor precio límite
- **AT price STOP** a nivel de arranque

Ejemplo AT MARKET

REM Compra de 100 títulos al precio del mercado

BUY 100 Shares AT MARKET

Ejemplo LIMIT

REM Compra de 100 títulos al precio límite de 15.45€

BUY 100 Shares AT 15.45 LIMIT

Ejemplo STOP

REM Compra de 100 títulos a nivel de arranque, el nivel siendo de 16.18€

BUY 100 Shares AT 16.18 STOP

Fecha de ejecución

Por defecto, cada orden se ejecuta en la barra que sigue. Sin embargo, en el caso de órdenes al precio del mercado, es posible avanzar o regresar la fecha de ejecución de las mismas haciendo uso de las instrucciones siguientes:

- **ThisBarOnClose** al cierre de la barra actual
- **NextBarOpen** a la apertura de la barra que sigue (**orden por defecto**)
- **NextBarClose** al cierre de la barra que sigue
-
- **TodayOnClose** al cierre del día de hoy (utilizado en intradía)
- **TomorrowOpen** a la apertura del día de mañana (utilizado en intradía)
- **TomorrowClose** al cierre del día de mañana (utilizado en intradía)

Ejemplo ThisBarOnClose

REM Compra de 100 títulos al cierre de la barra en curso
BUY 100 Shares AT MARKET ThisBarOnClose

Ejemplo NextBarClose

REM Compra de 100 títulos al cierre de la barra que sigue
BUY 100 Shares AT MARKET NextBarClose

Ejemplo TodayOnClose

REM Compra de 100 títulos al cierre del día de hoy (intradía)
BUY 100 Shares AT MARKET TodayOnClose

Ejemplo TomorrowOpen

REM Compra de 100 títulos a la apertura del día de mañana (intradía)
BUY 100 Shares AT MARKET TomorrowOpen

Ejemplo TomorrowClose

REM Compra de 100 títulos al cierre del día de mañana (intradía)
BUY 100 Shares AT MARKET TomorrowClose

Simulación de stop-loss gestionados por programación

Set Stop

SET STOP *price*

Esta instrucción le permite añadir un stop de acuerdo con la fórmula que usted mismo está gestionando. Note por favor que en la interfaz de ProBacktest puede usted simular 4 tipos de stops clásicos sin tener que programarlos.

Ejemplo Set Stop

REM Un stop de seguimiento

```
IF close > AVERAGE[30] AND NOT OnMarket THEN  
  BUY 100 Shares AT MARKET
```

REM Determina la separación entre el precio de ejecución efectivo y el precio ideal
Spread = **openOfNextBar - low**

REM Determina el precio +alto desde la entrada en el mercado
Alto = **openOfNextBar**

ENDIF

REM Stop de seguimiento que mantiene una distancia constante con el precio más alto

```
IF OnMarket THEN  
  Alto = MAX(Alto, high)  
  SET STOP (Alto - Spread)
```

ENDIF

Estrategias de backtest

AS

Tanto las instrucciones como las variables de ProBackTest están asociadas a una o varias estrategias. Se pueden definir tantas estrategias como se quiera.

Si usted no otorga un nombre particular a una determinada estrategia, ProBackTest le asigna uno implícitamente y todas sus órdenes quedan referidas a ella.

Si se definen varias estrategias a la vez, lo mejor es atribuirles un nombre. De esta forma, se asegura que cuando se aplica una instrucción o una variable a una determinada estrategia, esta instrucción o esta variable se aplica únicamente a la estrategia en cuestión. En caso contrario, la instrucción o la variable se aplica a todas las estrategias.

Para definir una estrategia, es suficiente de añadir, justo después de haber indicado su orden, la palabra clave **AS** seguida del nombre de su estrategia:

Ejemplo órdenes asociadas a una sola estrategia

```
REM Estrategia "media móvil"
```

```
IF close > AVERAGE[30](close) AND NOT OnMarket THEN  
    BUY 100 Shares AT MARKET AS « media movil »  
ENDIF
```

De esta manera, todas las consideraciones relativas a la posición en curso (líquido, corto, largo) se hacen dentro del contexto propio de la estrategia. Usted puede globalmente estar en corto, pero no desde el punto de vista de la estrategia en curso.

Veamos el caso de varias estrategias sin especificación:

Ejemplo órdenes asociadas a todas las estrategias

```
REM Orden emitida simultaneamente sobre todas las estrategias  
SELL 50 Shares AT MARKET
```

Esta orden repercute en efecto sobre todas las estrategias. Si usted tiene dos, dicha orden ejecutará la venta de $2 * 50$ títulos = **100 títulos**. Ello puede resultar útil a la hora de cerrar todas sus posiciones en curso emitiendo una sola instrucción.

Estos principios se aplican también para el tratamiento de las variables de backtesting, fuera de las instrucciones de compra y venta.

Ejemplo EntryIndex, estrategias

REM Estrategia "media móvil"

```
IF close > AVERAGE[30](close) AND NOT OnMarket THEN  
    BUY 100 Shares AT MARKET AS « media móvil »  
ENDIF
```

REM Venta en el corte del nivel más bajo de la barra de entrada

```
IF close < low[BarIndex - EntryIndex AS « media móvil »] THEN  
    SELL AT MARKET AS "media móvil"  
ENDIF
```

Variables de estado

OnMarket, LongOnMarket, ShortOnMarket

Estas variables indican en todo momento el estado de la cartera de backtest. Este puede ser líquido, comprador o vendedor (al descubierto).

Descripción

En general, a una señal dada, un sistema reacciona de manera diferente según que las posiciones en curso sean largas, cortas o líquidas. En efecto, vender al descubierto o vender una posición larga son dos cosas muy distintas: en un caso, se empieza una posición, en el otro se pone fin a una existente.

Las condiciones asociadas son también, en general, distintas. Por ejemplo, se empieza una posición cuando una señal nos indica el nacimiento de una tendencia. Recíprocamente, se pone fin a ella cuando aparece una señal contraria o cuando, por razones estratégicas, así nos lo indica el money management.

Ejemplo OnMarket

```
REM Compra por ruptura de la media móvil
```

```
Signal = close > AVERAGE[30](close)
```

```
REM La señal es suficiente para entrar en el mercado
```

```
IF NOT OnMarket THEN
```

```
    IF Signal THEN
```

```
        BUY 10 SHARES AT MARKET
```

```
    ENDIF
```

```
ENDIF
```

```
REM Tenemos dos señales de salida...
```

```
IF OnMarket THEN
```

```
    REM La línea de abajo genera una señal de salida por ruptura de un más bajo
```

```
    SignalSortie = close < LOWEST[10](low[1])
```

```
    REM La señal de salida es doble (no-simétrica en relación a la entrada)
```

```
    IF NOT Signal OR SignalSortie THEN
```

```
        SELL 10 SHARES AT MARKET
```

ENDIF

ENDIF

Variables de seguimiento de posición

CountOfLongShares, CountOfShortShares, CountOfPosition

Estas variables representan respectivamente:

- el número de títulos puestos en juego a la alza
- el número de títulos puestos en juego a la baja
- el número de órdenes acumuladas en un sentido determinado

Descripción

En un instante determinado, dichas variables ofrecen una información más precisa que la que ofrecen las variables de estado. No solamente permiten de saber si se está en el mercado sino que también contabilizan el número de títulos acumulados así como el número de órdenes.

Se pueden pues afinar las tomas de posición. No tan sólo se podrán indicar condiciones diferentes para la entrada y salida del mercado sino que además se podrán gestionar las acumulaciones (a la alza o a la baja).

Ejemplo CountOfLongShares, CountOfPosition

```
REM Compra por ruptura de media móvil
```

```
Signal = close > AVERAGE[30](close)
```

```
REM La señal es suficiente para entrar en el mercado
```

```
IF NOT OnMarket THEN
```

```
    IF Signal THEN
```

```
        BUY 10 SHARES AT MARKET
```

```
    ENDIF
```

```
ENDIF
```

```
REM Acumulación hasta 3 veces en tanto que la señal sea válida
```

```
IF OnMarket THEN
```

```
    REM La línea de abajo genera una señal de salida por ruptura de un más bajo
```

```
    SignalSortie = close < LOWEST[10](low[1])
```

```
    REM La señal de salida final es doble (no-simétrica en relación a la entrada)
```

```
    IF NOT Signal OR SignalSortie THEN
```

```
SELL CountOfLongShares SHARES AT MARKET
```

```
REM Acumulación de las posiciones hasta 3 veces si no hay señal de salida
```

```
ELSIF CountOfPosition < 3 THEN
```

```
    BUY 10 SHARES AT MARKET
```

```
ENDIF
```

```
ENDIF
```

Notas :

Para que la acumulación de órdenes sea posible, debe activarse la opción « *Acumular las posiciones* » de la gestión del capital.

Cuando se activa la opción « *1 stop para el conjunto de posiciones* », todas las posiciones se fusionan en una sola. En este caso, la variable **CountOfPosition** no puede exceder de 1.

Acceso a las barras relativas a la última orden

EntryIndex

Representa el índice de la barra sobre la cual se ejecutó la última orden.

Descripción

Este parámetro permite, por ejemplo, de explotar las características del candlestick sobre el cual se ha entrado en el mercado con el fin de ajustar los stops.

Ejemplo EntryIndex

```
REM Compra por ruptura de media móvil
IF NOT OnMarket THEN
    IF close > AVERAGE[30](close) THEN
        BUY 100 %CAPITAL AT MARKET
    ENDIF
ENDIF

REM Salida por ruptura del más bajo del candlestick sobre el cual se ha entrado
IF OnMarket THEN
    SELL AT low[BarIndex - EntryIndex] STOP
ENDIF
```

EntryQuote

Representa el precio de ejecución de la última orden.

Descripción

Este parámetro permite, por ejemplo, el cálculo de un stop basada en la cotización de entrada de la última orden simulada.

Exemple EntryQuote

```
REM Compra por ruptura de media móvil
IF NOT OnMarket THEN
    IF close > AVERAGE[30](close) THEN
        BUY 100 %CAPITAL AT MARKET
    ENDIF
ENDIF
```

```
REM Salida por ruptura del nivel de precio (curso) en el cual se ha realizado la orden de compra
IF OnMarket THEN
    SELL AT EntryQuote STOP
ENDIF
```

Ejemplos

PennyStocks

Descripción

Presentamos aquí un sistema largo (únicamente posiciones a la compra) con el objetivo de generar un mínimo de señales posibles y de aprovechar las mejores oportunidades.

Este dispositivo comporta pues un « filtro » de oportunidades basado en el valor absoluto de los precios. En este caso particular, el sistema se concentra en los llamados « penny stocks » (cuyo curso es inferior a 10€)

Las señales de entrada son bien conocidas: sobre un nuevo más alto cuando una media móvil confirma la tendencia.

La señal de salida también es clásica: ruptura de la media móvil para los más bajos a efectos de limitar las señales falsas.

Se trata efectivamente de aprovechar por completo las oportunidades que se presentan con los penny stocks (las cuales pueden presentarse en forma muy rápida y muy elevada), dejando correr los beneficios más elevados, a pesar de perder en ello una pequeña parte.

Ejemplo PennyStocks

```
REM CONDICIONES DE COMPRA
```

```
REM el +alto del día sobrepasa el +alto de los 10 días precedentes
```

```
ha1 = high > highest[10](high[1])
```

```
REM el +alto del día es superior a la MM13 (media móvil 13)
```

```
ha2 = high > exponentialAverage[13](close)
```

```
REM el curso es inferior a 10
```

```
ha3 = close < 10
```

```
REM el cierre es superior a la apertura
```

```
ha4 = close > open
```

```
IF ha1 AND ha2 AND ha3 AND ha4 THEN
```

```
    BUY 70 %capital AT MARKET
```

```
ENDIF
```

REM CONDICIÓN DE SALIDA

REM el +bajo cruza la MM30 de los más bajos a la baja

IF **low** CROSSES UNDER exponentialAverage[30](**low**) THEN

 SELL AT MARKET

ENDIF

Entrada optimizada

Descripción

Presentamos aquí un sistema largo (únicamente posiciones a la compra) con la condición de no entrar en el mercado que si se produce un repliegue significativo de la cotización dentro de una tendencia a la alta.

La tendencia se determina por la media móvil ponderada a 26 barras.

A fin de entrar en el « mejor » repliegue en términos de rentabilidad/riesgo, optamos aquí por el indicador « Super Trend » desarrollado por O.Seban. Se trata de un seguidor de tendencia adaptado a la volatilidad de los títulos. Este indicador nos permitirá de encontrar nuestros puntos de entrada.

En efecto, cuando el indicador está por debajo de las cotizaciones, nos confirma por un lado la tendencia al alta pero, por otro lado, nos indica hasta donde las cotizaciones pueden refluir sin que se invierta la tendencia.

Compraremos a uno de estos niveles (a precio límite). El riesgo es que se produzca la ruptura de este nivel. Sin embargo, con la condición sobre la media móvil, tenemos una buena base para introducirnos en la tendencia en curso. Por otra parte, no hay que olvidar lo que todos conocemos en lo que se refiere a la tendencia (es más probable de que persista que de que se invierta). En principio, se puede pues apostar en ello.

Por supuesto, cualquier ruptura de la media móvil invalidaría nuestro escenario, y saldríamos entonces sin tardar.

Ejemplo Entrada optimizada

```
REM seguidor de tendencia optimizada de O.Seban
seguidor = Supertrend[1, 27]

REM ¿es el cierre superior a la media móvil ponderada a 26 semanas?
ha1 = close > weightedaverage[26](close)

REM ¿está el seguidor de O.Seban por debajo de las cotizaciones?
ha2 = close > seguidor

IF ha1 THEN
  IF ha2 THEN
    BUY 80 %capital AT seguidor LIMIT
  ENDF
ENDIF
```

```
ELSE
```

```
    SELL AT MARKET
```

```
ENDIF
```

Tendencia detectada sobre las cotizaciones

Descripción

Consideramos aquí un sistema largo/corto que está solicitado en forma permanente, sea a la compra, sea a la venta al descubierto.

El objetivo que nos proponemos es el de invertir la posición para aprovechar los movimientos de pánico ya sea a la compra, ya sea a la venta.

Nos basamos en las cotizaciones para detectar estos movimientos. La variación de los precios indica la tendencia. Se supone que toda variación más fuerte que la media es un signo precursor de la tendencia que va a seguir.

El money management es aquí bastante sencillo: 70% de capital al alza, y 40% a la baja. Se toman menos riesgos cuando se juega a la baja puesto que, en general, los movimientos son más devastadores, lo cual permite la obtención de ganancias más rápidas y más importantes cuando se tiene razón. Ello evita también el encontrarse en una situación demasiado peligrosa en caso de movimiento adverso (la pérdida potencial en el caso de una venta al descubierto es teóricamente ilimitada)

Ejemplo Tendencia detectada sobre las cotizaciones

REM determina la variación media de las cotizaciones de un día al otro

```
avt = AverageTrueRange[20](close)
```

REM criterio de compra : el alza es más intensa que la variación media (=pánico a la compra)

```
IF close > close[1] + avt THEN
```

```
    BUY 70 %capital AT MARKET
```

```
ENDIF
```

REM criterio de venta : la baja es más intensa que la variación media (=pánico a la venta)

```
IF close < close[1] - avt THEN
```

```
    SELLSHORT 40 %capital AT MARKET
```

```
ENDIF
```

Sell in may and go away !

Descripción

Se pueden también concebir sistemas que hacen abstracción de la gráfica de cotizaciones y que se basan más bien en datos estadísticos.

Se sabe que el proverbio « Sell in may and go away ! » ha nacido de una anomalía estadística. En efecto, un estudio generalizado de los mercados hecho a partir de históricos relativamente extensos muestra que aquéllos realizan sus peores resultados entre los meses de mayo y septiembre, ambos inclusivos.

El objetivo aquí es aprovechar este resultado con la ayuda de un sistema de trading más o menos complejo.

Dicho sistema puede funcionar tanto en « daily », como en « weekly » (los resultados son casi idénticos)

Ofrece una « equity curve » de una calidad muy aceptable puesto que se obliga a alternar compras y ventas al descubierto cualesquiera que sean las trampas tendidas por el mercado (y estadísticamente con un buen timing). Ello limita el efecto de la contra-tendencia demasiado prolongada.

Ejemplo Tendencia detectada sobre los meses (!!)

REM de acuerdo con el título.... Bien, vamos a vender en mayo !

IF Month = 5 THEN

SELLSHORT 50%capital AT MARKET

REM no está en el título pero el estudio considera que, en general, se toca el fondo durante el mes de octubre

REM ... y con las fiestas de fin de año, la bolsa no puede seguir cayendo !

ELSIF Month = 10 THEN

BUY 50%capital AT MARKET

ENDIF

Intradía 2 barras

Descripción

Veamos ahora un ejemplo célebre de sistema intradía. Dicho sistema consiste en operar sobre las rupturas de soportes/resistencias determinados durante las dos primeras barras de la jornada. La salida se efectúa al cierre del día.

Ejemplo Intradía 2 barras

```
REM Cierre de la segunda barra (indice 1)
IF intradíaBarIndex = 1 THEN

    plusHaut = Highest[2](high)
    plusBas = Lowest[2](low)

ENDIF

REM Compra/venta sobre ruptura entre la 3era barra y las 16:00:00 (hora local)
IF intradíaBarIndex > 1 AND Time < 160000 THEN

    REM Ruptura de resistencia
    IF close > plusHaut THEN
        BUY 70%capital AT MARKET
        SELL AT MARKET TodayOnClose

    REM Ruptura de soporte
    ELSIF close < plusBas THEN
        SELLSHORT 70%capital AT MARKET
        EXITSHORT AT MARKET TodayOnClose

    ENDIF

ENDIF
```